
Каждый язык обладает структурой, называемой грамматикой, или синтаксисом.
Например, правильное предложение (в английском языке – Прим. перев.)  всегда
состоит из подлежащего и следующего за ним сказуемого. Под правильным здесь
понимается правильно составленное предложение. Это можно описать следующей
формулой:

��������	
� = ���������� ��������.

Если мы добавим к этой формуле еще две

����������  = "���	" | "���
�".
��������  = "��" | "�����
�".

то с их помощью получим  ровно четыре возможных предложения, а именно:

���	 �� ���
� ��
���	 �����
� ���
� �����
�

где символ | должен произноситься как или. Мы назовем эти формулы синтакси�
ческими правилами, продукциями или просто синтаксическими уравнениями. ���-
������� и �������� – это синтаксические классы. Краткая запись этих формул
пренебрегает смыслом идентификаторов:

S = AB. L = {ac, ad, bc, bd}
A = "a" | "b".
B = "c" | "d".

Мы будем использовать такую сокращенную запись в последующих кратких
примерах. Множество L предложений, которые могут быть сгенерированы этим
способом, то есть повторяющейся заменой левых частей уравнений правыми, на&
зывается языком.

Приведенный выше пример, очевидно, определяет язык, состоящий только из
четырех предложений. Обычно язык содержит бесконечно много предложений.
Следующий пример показывает, что бесконечное множество может быть очень
просто определено конечным числом уравнений. Символ � обозначает пустую
последовательность.

S = A. L = {�, a, aa, aaa, aaaa,...}
� = "a" � | �.

Метод, позволяющий выполнять подстановку (здесь "a"A вместо �) бесконеч&
ное число раз, называется  рекурсией.

Наш третий пример опять основан на применении рекурсии. Но он генерирует
не только предложения, состоящие из произвольной последовательности одного
и того же символа, но и вложенные предложения:

S = A. L = {b, abc, aabcc, aaabccc,...}
A = "a" A "c" | "b".

Понятно, что таким образом может быть выражена произвольно глубокая вло&
женность (здесь – для  A), что особенно важно в определении структурированных
языков.

Наш четвертый, и последний, пример показывает структуру выражений. Сим&
волы E, T, F и V обозначают выражение, слагаемое, множитель и переменную соот&
ветственно.

E = T | "+" T.
T = F | T "*" F.
F = V | "(" E ")".
V = "a" | "b" | "c" | "d".

Из этого примера видно, что синтаксис не только определяет множество пред&
ложений языка, но и наделяет их структурой. Синтаксис раскладывает предложе&
ния на составляющие, как показано в примере на рис. 2.1. Графические представ&
ления называются структурными деревьями, или синтаксическими деревьями.

Рис. 2.1. Структура выражений

Сформулируем представленные выше понятия более строго.
Язык (порождающая язык грамматика – Прим. перев.) определяется следую&

щим образом:

1. Множество терминальных символов. Это символы, которые появляются
в предложениях. Говорят, что они терминальные, потому что не могут быть
заменены никакими другими символами. Процесс подстановки заканчива&
ется терминальными символами. В нашем первом примере это множество
состоит из элементов a, b, c и d. Это множество также называется словарем.

2. Множество нетерминальных символов. Они обозначают синтаксические
классы и могут замещаться в результате подстановок. В нашем первом при&
мере это множество состоит из элементов S, A и B.

3. Множество синтаксических уравнений (также называемых продукциями).
Они определяют возможные подстановки нетерминальных символов.
Уравнение задается для каждого нетерминального символа.

Язык и синтаксис

4. Начальный символ. Это нетерминальный символ, обозначаемый в примерах
как S.

Таким образом, язык – это множество цепочек терминальных символов, кото&
рые могут быть выведены из начального символа многократным применением
синтаксических уравнений, то есть подстановок.

Желательно также строго и точно определить нотацию, в которой записывают&
ся синтаксические уравнения. Пусть нетерминальные символы будут идентифи&
каторами, как в языках программирования, то есть последовательностями букв
(и, возможно, цифр), например expression, term. Пусть терминальные символы
будут последовательностями символов, заключенными в кавычки (строками), на&
пример "=", "|". Для определения структуры этих уравнений удобно воспользо&
ваться тем же самым инструментом, который только что был определен:

syntax = production syntax | �.
production = identifier "=" expression ".".
expression = term | expression "|" term.
term = factor | term factor.
factor = identifier | string.
identifier = letter | identifier letter | identifier digit.
string = stringhead""".
stringhead = """ | stringhead character.
letter = "A" | ... | "Z".
digit = "0" | ... | "9".

Эта нотация почти в таком же виде была введена в 1960 году Дж. Бэкусом и П.
Науром для формального описания синтаксиса языка Алгол 60 и поэтому полу&
чила название формы Бэкуса–Наура (БНФ) [12]. Как показывает пример, ис&
пользование рекурсии для простых повторений несколько мешает их восприя&
тию. Поэтому мы расширим эту нотацию двумя конструкциями, выражающими
повторение и необязательность. Кроме этого, разрешим выражения заключать в
скобки. Таким образом, вводится расширение БНФ, называемое РБНФ [17], ко&
торым мы снова воспользуемся для его же точного определения:

syntax = { production }.
production = identifier "=" expression ".".
expression = term {"|" term }.
term = factor { factor }.
factor = identifier | ����� | "(" expression ")" | "[" expression "]"

   | "{" expression ”}".
identifier = letter { letter | digit }.
string = """ {character} """.
letter = "A" |... | "Z".
digit = "0" |... | "9".

Множитель вида {x} равнозначен произвольно длинной последовательности x,
включая пустую последовательность. Продукция вида

� = AB | �.

теперь записывается короче:  � = {B}. Множитель вида [x] равнозначен «x, или нич&
то», то есть выражает необязательность. Следовательно, потребность в специаль&
ном символе � для пустой цепочки исчезает.

Идея определять языки и их грамматику с математической точностью восхо&
дит к Н. Хомскому (N. Chomsky). Однако стало ясно, что предложенная простая
схема правил подстановки недостаточна для представления всей сложности раз&
говорных языков. Положение не изменилось даже после того, как формализм был
значительно расширен. Но зато эта работа оказалась чрезвычайно плодотворной
для теории языков программирования и математических формализмов. С его по&
мощью Алгол 60 стал первым языком программирования, который был определен
точно и формально. Мимоходом подчеркнем, что эта точность относилась только
к синтаксису, но не к семантике.

Термин язык программирования также обязан формализму Хомского, по&
скольку языки программирования, оказывается, обладают структурой, подобной
структуре разговорных языков. Но мы уверены, что этот термин, в общем, доволь&
но неудачен, поскольку на языке программирования нельзя разговаривать, и по&
этому он не язык в прямом смысле слова. Более подходящими были бы термины
формализм или формальная нотация.

Некоторые удивляются, почему точному определению предложений языка
должно придаваться такое большое значение; ведь в действительности это не так.
Тем не менее очень важно, чтобы предложение было правильно составлено. Хотя
и в этом случае предложение может потребовать уточнения. Но, в конце концов,
структура предложения (правильно составленного) важна потому, что является
инструментом понимания его смысла. Благодаря синтаксической структуре от&
дельные составные части предложения и их смысл могут распознаваться незави&
симо, а все вместе они придают смысл целому.

Давайте проиллюстрируем этот момент, используя следующий простой пример
выражения со сложением. Обозначим идентификатором E выражение, а N – число:

E = N | E "+" E.
N = "1" | "2" | "3" | "4".

Очевидно, "4 + 2 + 1" – правильное выражение, которое можно получить не&
сколькими способами, каждому из которых соответствует своя структура, как по&
казано на рис. 2.2.

Обе структуры могут быть представлены  скобочными выражениями, а имен&
но (4 + 2) + 1 и 4 + (2 + 1) соответственно. К счастью, благодаря ассоциативности
сложения результат в обоих случаях один и тот же и равен 7. Однако это не всегда
так. Если в нашем примере знак сложения заменить на знак вычитания, то две
структуры дадут разные результаты: (4 – 2) – 1 = 1, 4 – (2 – 1) = 3.

Пример иллюстрирует два факта:

1. Интерпретация предложений всегда основывается на распознавании син&
таксической структуры.

2. Каждое предложение должно иметь единственную структуру, для того что&
бы не быть двусмысленным.

Язык и синтаксис

Если второе требование не выполняется, может возникнуть двусмысленное
предложение. Этим можно обогатить разговорный язык; однако двусмысленные
языки программирования просто бесполезны.

Мы называем синтаксический класс неоднозначным, если ему можно соотнес&
ти несколько структур. Язык неоднозначен, если в него входит по крайней мере
один неоднозначный синтаксический класс (конструкция).

2.1. Упражнения
2.1. Сообщение об Алголе 60 содержит следующий синтаксис (приведенный
к РБНФ):

primary = unsignedNumber | variable | "(" arithmeticExpression ")" |... .
factor = primary | factor "�" primary.
term = factor | term ("*" | "/" | "�") factor.
simpleArithmeticExpression = term | ("+" | "–") term

    | simpleArithmeticExpression ("+" | "–") term.
arithmeticExpression = simpleArithmeticExpression |

"IF" BooleanExpression "THEN" simpleArithmeticExpression
"ELSE" arithmeticExpression.

relationalOperator = "=" | "�"| "�" | "<" | "�" | ">".
relation = arithmeticExpression relationalOperator arithmeticExpression.
BooleanPrimary = logicalValue | variable | relation

   | "("BooleanExpression")" |... .
BooleanSecondary = BooleanPrimary | "¬" BooleanPrimary.
BooleanFactor = BooleanSecondary | BooleanFactor "	" BooleanSecondary.
BooleanTerm = BooleanFactor | BooleanTerm "
" BooleanFactor.
implication = BooleanTerm | implication "=>" BooleanTerm.
simpleBoolean = implication | simpleBoolean "�" implication.
BooleanExpression = simpleBoolean |

"IF" BooleanExpression "THEN" simpleBoolean "ELSE" BooleanExpression.

Определите синтаксические деревья следующих выражений, в которых буквы
обозначают переменные:

Рис. 2.2. Разные структурные деревья для одного и того же выражения



25

x + y + z
x * y + z
x + y * z
(x – y) * (x + y)
–x � y
a + b < c + d
a + b < c VVVVV d � e 	  ¬f => g>h ? i *j = k � L VVVVV m–n + p � q

2.2. Следующие продукции также являются частью первоначального опреде&
ления Алгола 60. Они содержат двусмысленности, которые были устранены в
Пересмотренном сообщении.

forListElement = arithmeticExpression
 | arithmeticExpression "STEP" arithmeticExpression "UNTIL" arithmeticExpression
 | arithmeticExpression "WHILE" BooleanExpression .
forList = forListElement | forList "," forListElement .
forClause = "FOR" variable ":=" forList "DO" .
forStatement = forClause statement .
compoundTail = statement "END" | statement ";" compoundTail .
compoundStatement = "BEGIN" compoundTail .
unconditionalStatement = basicStatement | forStatement | compoundStatement|...
ifStatement = "IF" BooleanExpression "THEN" unconditionalStatement.
conditionalStatement = ifStatement | ifStatement "ELSE" statement .
statement = unconditionalStatement | conditionalStatement .

Найдите по крайней мере две различные структуры для следующих выраже&
ний и операторов. Пусть A и B обозначают «простые операторы».

IF a THEN b ELSE c = d
IF a THEN IF b THEN A ELSE B
IF  a  THEN  FOR ... DO  IF  b  THEN  A  ELSE   B

Предложите альтернативный однозначный синтаксис.
2.3. Рассмотрите следующие конструкции и попытайтесь разобраться, какие

из них являются корректными для Алгола, а какие – для Оберона (см. приложе&
ние 2):

a+b=c+d
a * –b
a<b  &  c<d

Вычислите следующие выражения:

5 * 13 DIV 4 =
13  DIV  5 *4 =

Упражнения

Синтаксические уравнения в том виде, как они определены в РБНФ, генерируют
контекстно&свободные языки. Термин «контекстно&свободный» принадлежит
Хомскому и происходит из того факта, что замена символа слева от знака  =  це&
почкой, порожденной выражением справа от знака =, возможна всегда, независи&
мо от контекста этого символа внутри предложения. Оказывается, что эта свобода
контекста (по Хомскому) очень подходит и даже желательна для языков програм&
мирования. Но контекстная зависимость в ином смысле все&таки необходима. Мы
вернемся к этой теме в главе 8.

Здесь мы прежде всего хотим исследовать подкласс, а не контекстно&свобод&
ные языки вообще. Этот подкласс, известный как регулярные языки, играет суще&
ственную роль в области языков программирования. В сущности, это контекстно&
свободные языки, синтаксис которых не содержит иной рекурсии, кроме
спецификации повторения. Так как в РБНФ повторение определено непосред&
ственно и без использования рекурсии,  можно дать простое определение:

Язык регулярен, если его синтаксис может быть выражен единствен&
ным выражением РБНФ.

Требование достаточности единственного уравнения подразумевает, что вы&
ражение состоит только из терминальных символов. Такое выражение называют
регулярным выражением.

Видимо, достаточно двух кратких примеров регулярных языков. Первый оп&
ределяет идентификаторы, поскольку они встречаются в большинстве языков,
а второй – целые числа в десятичной записи. Для краткости мы используем нетер&
минальные символы letter и digit. Их можно исключить подстановкой, в результа&
те чего и identifier, и integer будут регулярными выражениями.

identifier = letter {letter | digit}.
integer = digit {digit}.
letter = "A" | "B" | ... | "Z"
digit = "0" |"1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9".

Причина нашего интереса к регулярным языкам заключается в том, что про&
граммы для распознавания регулярных выражений очень просты и эффективны.
Под «распознаванием» мы подразумеваем определение структуры предложения
и, следовательно, его правильность, то есть принадлежность языку. Распознава&
ние выражения называется синтаксическим анализом.

Для распознавания регулярных выражений необходимо и достаточно иметь
конечный автомат, также называемый машиной состояний (state machine). На
каждом шаге машина состояний читает следующий символ и меняет свое состоя&
ние. Следующее состояние определяется исключительно предыдущим состояни&
ем и очередным прочитанным символом. Если очередное состояние уникально, то
машина состояний является детерминированной, иначе недетерминированной.
Если машина состояний реализована программно, то состояние машины опреде&
ляется текущей точкой выполнения программы.

Анализирующая программа может быть выведена непосредственно из опреде&
ления синтаксиса в РБНФ. Для каждой РБНФ&конструкции KKKKK существует прави&

ло перевода, которое порождает фрагмент программы Pr(K)Pr(K)Pr(K)Pr(K)Pr(K). Правила перевода из
РБНФ в текст программы показаны ниже. В них sym – глобальная переменная,
представляющая собой последний символ исходного текста, прочитанный проце&
дурой next. Процедура error завершает выполнение программы, сигнализируя
о том, что последовательность символов не принадлежит языку.

KKKKK Pr(K)Pr(K)Pr(K)Pr(K)Pr(K)
"x" IF sym = "x" THEN next ELSE error END
(exp) Pr(exp)
[exp] IF sym IN first(exp) THEN Pr(exp) END
{exp} WHILE sym IN first(exp) DO Pr(exp) END
fac0 fac1 ... facN Pr(fac0); Pr(fac1); ... Pr(facN);
term0|term1|...|termN CASE sym OF

  first(term0): Pr(term0);
| first(term1): Pr(term1);
...
| first(termN): Pr(termN);
END

Множество first(KKKKK) содержит все символы, с которых могут начинаться пред&
ложения, полученные из конструкции KKKKK. Это множество начальных символов кон&
струкции KKKKK. Для двух примеров –  идентификаторов и целых чисел – они таковы:

first(integer) = digits = {"0", "1", "2", "3", "4", "5", "6", "7", "8", "9"}
first(identifier) = letters = {"A", "B",..., "Z"}

Однако применение этих простых правил перевода, генерирующих синтакси&
ческий анализатор по заданному синтаксису, возможно лишь при условии детер&
минированного синтаксиса. Это предварительное условие может быть сформули&
ровано более конкретно следующим образом:

KKKKK Cond(K)Cond(K)Cond(K)Cond(K)Cond(K)
term0 | terml Символы  term0 и term1 не должны иметь никаких общих

начальных символов
fac0 fac1 Если fac0 содержит пустую последовательность, то fac0 и

fac1 не должны иметь никаких общих начальных символов
[exp] 
�
 {exp} Множества начальных символов exp и символов, которые

могут следовать за KKKKK, не должны пересекаться

Очевидно, эти условия выполняются  в примерах для идентификаторов и це&
лых чисел, и мы, таким образом, получаем следующие программы для их распоз&
навания:

IF s IN letters THEN next ELSE error END;
WHILE sym IN letters + digits DO

CASE sym OF
  "A" .. "Z": next
| "0" .. "9": next
END

END

Регулярные языки

IF sym IN digits THEN next ELSE error END;
WHILE sym IN digits DO next END

Обычно программу, полученную применением правил перевода, можно упрос&
тить путем устранения условий, которые уже явно установлены предыдущими ус&
ловиями. Условия sym IN letters и sym IN digits записываются следующим образом:

("A" <= sym) & (sym <= "Z")
("0" <= sym) & (sym <= "9")

Важность регулярных языков для языков программирования следует из того,
что последние обычно определяются в два этапа. На первом этапе их синтаксис
определяется в терминах словаря абстрактных терминальных символов. На вто&
ром эти абстрактные символы определяются как цепочки конкретных терминаль&
ных символов, таких как ASCII&символы. Это второе определение обычно имеет
регулярный синтаксис. Разделение на два этапа обладает тем преимуществом, что
определение абстрактных символов и, таким образом, языка не зависит от конк&
ретного представления, то есть от некоторого конкретного набора символов, ис&
пользуемого на некотором конкретном оборудовании.

Такое разделение оказывает влияние и на структуру компилятора. Процесс
синтаксического анализа использует процедуру получения следующего абстракт&
ного символа, а эта процедура, в свою очередь, распознает конкретные символы –
цепочки из одной или более литер. Эту последнюю процедуру называют лекси&
ческим анализатором, а синтаксический анализ на этом втором, более низком
уровне – лексическим анализом. Определение символов, выражаемых литерами,
обычно задается в форме регулярного языка, и поэтому лексический анализатор –
типичная  машина состояний.

Суммируем различия между этими двумя уровнями следующим образом:

ПроцессПроцессПроцессПроцессПроцесс ВходнойВходнойВходнойВходнойВходной АлгоритмАлгоритмАлгоритмАлгоритмАлгоритм СинтаксисСинтаксисСинтаксисСинтаксисСинтаксис
элементэлементэлементэлементэлемент

Лексический анализ Литера Лексический анализатор Регулярный
Синтаксический анализ Символ Синтаксический анализатор Контекстно%свободный

В качестве примера приведем лексический анализатор для разбора РБНФ. Его
терминальные символы и их определение в терминах литер таковы:

symbol = {blank} (identifier | string | "(" | ")" | "[" | "]"
  | "{" | "}" | "|" | "=" | ".") .

identifier = letter {letter | digit}.
string = """ {character} """.

Из этих определений мы выведем процедуру GetSym, которая при каждом вы&
зове присваивает глобальной переменной sym числовое значение, представляю&
щее следующий прочитанный символ. Если символ является идентификатором
(identifier) или строкой (string), то конкретная цепочка литер присваивается до&
полнительной глобальной переменной id. Также отметим, что лексический анали&

затор обычно подразумевает следующее правило о пробелах и концах строк: про&
белы и концы строк разделяют цепочки литер и другого назначения не имеют.
Процедура GetSym на Обероне использует следующие объявления:

CONST IdLen = 32;
ident = 0; literal = 2; lparen = 3; lbrak = 4; lbrace = 5; bar = 6;
eql = 7; rparen = 8; rbrak = 9; rbrace= 10; period = 11; other = 12;

TYPE Identifier = ARRAY IdLen OF CHAR;

VAR ch: CHAR;
sym: INTEGER;
id: Identifier;
R: Texts.Reader;

Отметим, что абстрактная операция чтения теперь представлена конкретным
вызовом Texts.Read(R,ch). R – это глобально объявленный объект типа Reader, яв&
ляющийся источником исходного текста. Отметим также, что переменная ch дол&
жна быть глобальной, потому что по завершении GetSym она может содержать
первую литеру следующего символа. Это нужно иметь в виду при очередном об&
ращении к GetSym.

PROCEDURE GetSym;
VAR i: INTEGER;

BEGIN (*������ ��������*)
WHILE –R.eot & (ch <= " ") DO Texts.Read(R, ch) END ;
CASE ch OF
"A".. "Z", "a" .. "z":

sym := ident; i := 0;
REPEAT id[i] := ch; INC(i);

   Texts.Read(R, ch)
UNTIL (CAP(ch) < "A") OR (CAP(ch) > "Z");
id[i] := OX

| 22X: (*�������*)
Texts.Read(R, ch); sym := literal; i := 0;
WHILE (ch # 22X) & (ch > "") DO

id[i] := ch; INC(i); Texts.Read(R, ch)
END;
IF ch <= " " THEN error(l) END ;
id[i] := OX; Texts.Read(R, ch)

| "=": sym := eql; Texts.ReadfR, ch)
| "(" : sym := lparen; Texts.Read(R, ch)
| ")" : sym := rparen; Texts.Read(R, ch)
| "[" : sym := lbrak; Texts.Read(R, ch)
| "]" : sym := rbrak; Texts.Read(R, ch)
| "{" : sym := lbrace; Texts.Read(R, ch)
| "}" : sym := rbrace; Texts.Read(R, ch)
| "|" : sym := bar; Texts.Read(R, ch)
| ".": sym := period; Texts.Read(R, ch)

Регулярные языки

ELSE sym := other; Texts.Read(R, ch)
END

END GetSym

3.1. Упражнение
Предложения регулярных языков могут быть распознаны конечными автомата&
ми. Они обычно описываются диаграммами переходов. Каждый узел представля&
ет состояние, а каждая стрелка – переход из одного состояния в другое. Стрелка
помечена символом, который прочитан непосредственно перед переходом. Рас&
смотрите следующие диаграммы и опишите синтаксис соответствующих языков
в РБНФ.




